

**NASA TECHNICAL
MEMORANDUM**



NASA TM X-1340

NASA TM X-1340

FACILITY FORM 602

(ACCESSION NUMBER)
43
(PAGES)
TMX 1340
(NASA CR OR TMX OR AD NUMBER)

(THRU)
1
(CODE)
08
(CATEGORY)

**COMPUTER PROGRAM FOR
DESIGN-POINT PERFORMANCE
OF TURBOJET AND TURBOFAN
ENGINE CYCLES**

*by Michael R. Vanco
Lewis Research Center
Cleveland, Ohio*

GPO PRICE \$ _____

CFSTI PRICE(S) \$ _____

Hard copy (HC) _____

Microfiche (MF) _____

4-457 (10-65)

COMPUTER PROGRAM FOR DESIGN-POINT PERFORMANCE OF
TURBOJET AND TURBOFAN ENGINE CYCLES

By Michael R. Vanco

Lewis Research Center
Cleveland, Ohio

NATIONAL AERONAUTICS AND SPACE ADMINISTRATION

For sale by the Clearinghouse for Federal Scientific and Technical Information
Springfield, Virginia 22151 - Price \$2.00

COMPUTER PROGRAM FOR DESIGN-POINT PERFORMANCE OF TURBOJET AND TURBOFAN ENGINE CYCLES

by Michael R. Vanco
Lewis Research Center

SUMMARY

The FORTRAN 7094 computer program for the calculation of the design-point performance of turbojet and turbofan engine cycles is presented along with the thermodynamic equations used. This program requires as input the airplane Mach number, the altitude-state conditions, turbine-inlet temperature, afterburner temperature, duct-burner temperature, bypass ratio, coolant flow, component efficiencies, and component pressure ratios. The output yields specific thrust, specific fuel consumption, engine efficiency, and several component temperatures and pressures. The thermodynamic properties of the gas are expressed as functions of temperature and fuel to air ratio. Sample input and output are included for an example case.

INTRODUCTION

Advanced aircraft for supersonic flight, high-payload long-range subsonic flight, and vertical flight require the development of advanced propulsion systems. The study of system characteristics and components for such airbreathing propulsion systems has been undertaken at the Lewis Research Center. As part of these studies, thermodynamic analyses of design-point performance of turbojet and turbofan engine cycles must be performed. To assist in these analyses, a computer program was developed.

The program requires as input the airplane Mach number, the altitude-state conditions, turbine-inlet temperature, afterburner temperature, duct-burner temperature, bypass ratio, coolant flow, component efficiencies, and component pressure ratios. The thermodynamic properties used in this analysis are expressed as functions of temperature and fuel to air ratio. The fuel is assumed to be of the form $(CH_2)_n$. The results of the analysis include specific thrust, specific fuel consumption, engine efficiency, and several component temperatures and pressures.

The equations used in the analysis and the FORTRAN 7094 computer program are presented in this report along with a description of the input and output. Sample input and output are included for an example case.

METHOD OF ANALYSIS

An analysis for the performance of a general engine is derived herein. The performance of this engine can be studied in terms of specific fuel consumption and specific thrust. The general engine consists of components of turbojet and turbofan engines. An analysis of the performance of these specific engines is obtained from the general engine analysis by proper specification of the input.

Cycle Description

The general engine cycle is shown in figure 1. The air enters the diffuser and the major portion of its velocity head is changed into a pressure head. The lower velocity air then enters the fan and is compressed. The air flow is then divided into the main flow and the duct flow. The main flow enters the compressor and is further compressed. The major portion of the main flow then enters the combustor with fuel, and combustion occurs. The small remaining portion of the main flow bypasses the combustor and is used to cool the turbine. The combustor exit gases are then expanded in the turbine, producing work to drive the compressor and fan. The turbine exit gases and the coolant flow then mix and enter the afterburner with fuel, and combustion occurs. The hot gases are then expanded in the main nozzle to produce thrust. The duct flow enters the duct-burner with fuel, and combustion occurs. The hot duct gases are then expanded in a nozzle to produce additional thrust.




Figure 1. - Schematic diagram of general engine.

Derivation of Equations

The equations used in the analysis of the general engine are derived in this section. The thermodynamic properties used for this analysis are functions of temperature and fuel to air ratio. The specific heat of the gas is expressed by a polynomial equation. Appropriate integrations of this equation yield the enthalpy change and the entropy function. The entropy function is herein defined as

$$\Delta\phi = \int_{T_1}^{T_2} \frac{C_p}{T} dT$$

(All symbols are defined in appendix A.) The derivation of the equations for specific heat, enthalpy change, and entropy function are given in appendix B. Since the specific heat is a function of temperature and fuel to air ratio, it is expressed as $C_p(T, f)$ in this analysis. Since

$$\Delta h = \int_{T_1}^{T_2} C_p(T, f) dT$$

and

$$\Delta\phi = \int_{T_1}^{T_2} \frac{C_p(T, f)}{T} dT$$

they are expressed as $\Delta h(T_2, T_1, f)$ and $\Delta\phi(T_2, T_1, f)$, respectively. When the fuel to air ratio is zero, these quantities will appear with temperatures only. The fuel is assumed to be of the form $(CH_2)_n$.

Engine inlet. - The static inlet conditions of the diffuser T_0 and p_0 are a function of the altitude. The inlet velocity is

$$V_0 = M_0 \sqrt{g R_a \gamma T_0} \quad (1)$$

where M_0 is the Mach number at which the airplane is traveling. The specific heat ratio is

$$\gamma = \frac{1}{1 - \frac{R_a}{C_p(T_0)J}} \quad (2)$$

The total temperature at the inlet is

$$T'_0 = T_0 + \frac{V_0^2}{2gJC_p} \quad (3a)$$

where

$$C_p = \frac{\Delta h(T'_0, T_0)}{T'_0 - T_0} \quad (3b)$$

The correct total temperature is then obtained by an iterative procedure involving equations (1) to (3b). The total pressure at the diffuser inlet is obtained from

$$\frac{R_a}{J} \ln \frac{P'_0}{P_0} = \Delta \varphi(T'_0, T_0)$$

Therefore,

$$P'_0 = P_0 e^{\Delta \varphi(T'_0, T_0)J/R_a} \quad (4)$$

Diffuser. - Since there is adiabatic flow in the diffuser,

$$T'_1 = T'_0 \quad (5)$$

The pressure ratio across the diffuser $r_{1,0}$ is an input parameter, which is a function of the Mach number. Therefore,

$$\frac{P'_1}{P'_0} = r_{1,0} \quad (6)$$

An example variation of this parameter is presented in reference 1.

Fan. - The fan pressure ratio P'_1/P'_1 is a variable. To determine the ideal fan-

exit temperature, the isentropic flow equation is used. Therefore,

$$\Delta\phi_F = \frac{R_a}{J} \ln \frac{P'_{1'}}{P'_1} \quad (7a)$$

and

$$\Delta\phi(T'_{1'}, id, T'_1) = \Delta\phi_F \quad (7b)$$

Therefore, $T'_{1'}, id$ can be determined from equation (7b). The fan work is

$$\Delta h_F = \frac{\Delta h(T'_{1'}, id, T'_1)}{\eta_F} \quad (8a)$$

where fan efficiency is a design parameter. The fan-exit temperature T'_1 is determined from

$$\Delta h(T'_{1'}, T'_1) = \Delta h_F \quad (8b)$$

The total airflow is

$$W_{tot} = W_{a,D} + W_{a,m} \quad (9a)$$

where $W_{a,D}$ is the duct airflow and $W_{a,m}$ is the main airflow. The ratio of the duct airflow to the main airflow $W_{a,D}/W_{a,m}$ is called the bypass ratio b . Therefore,

$$W_{tot} = (1 + b)W_{a,m} \quad (9b)$$

Compressor. - The overall pressure ratio of the fan and compressor is a variable. Thus, the compressor pressure ratio is

$$\frac{P'_2}{P'_{1'}} = \frac{P'_2/P'_1}{P'_{1'}/P'_1} \quad (10)$$

The ideal compressor-exit temperature $T'_{2, id}$ is obtained from

$$\Delta\phi(T'_{2, id}, T'_{1'}) = \frac{R_a}{J} \ln \frac{P'_2}{P'_{1'}} \quad (11)$$

The compressor work is

$$\Delta h_C = \frac{\Delta h(T'_{2, id}, T'_1)}{\eta_C} \quad (12a)$$

where the compressor efficiency is a design parameter. The compressor exit temperature T'_2 is determined from

$$\Delta h(T'_2, T'_1) = \Delta h_C \quad (12b)$$

Combustor. - An energy balance for the combustor yields

$$W_{f, m} h_f + \eta_B W_f HVF + (W_{a, m} - W_c) h_a = (W_{a, m} - W_c + W_f) h_g \quad (13)$$

For the enthalpy of the fuel to be equal to zero, the enthalpy reference temperature T_R must be equal to the temperature of the incoming fuel. As discussed in appendix B, the enthalpy change of the gas can be expressed as

$$\Delta h_g = (\Delta h_a + \Delta h_b f) \frac{1}{1 + f} \quad (14)$$

Therefore, substituting equation (14) into equation (13) and dividing by $(W_{a, m} - W_c)$ yield

$$\eta_B \frac{W_{f, m} HVF}{W_{a, m} - W_c} + \Delta h(T'_2, T_R) = \Delta h(T'_3, T_R) + \frac{W_{f, m}}{W_{a, m} - W_c} \Delta h_b(T'_3, T_R)$$

The fuel to air ratio based on combustor airflow is

$$\frac{W_{f, m}}{W_{a, m} - W_c} = \frac{\Delta h(T'_3, T_R) - \Delta h(T'_2, T_R)}{HVF \eta_B - \Delta h_b(T'_3, T_R)} \quad (15)$$

where the turbine inlet temperature T'_3 and the combustor efficiency η_B are design parameters. The fuel to air ratio based on main flow is

$$\frac{W_{f, m}}{W_{a, m}} = \frac{W_{f, m}}{W_{a, m} - W_c} \left(\frac{W_{a, m} - W_c}{W_{a, m}} \right) = \frac{W_{f, m}}{W_{a, m} - W_c} \left(1 - \frac{W_c}{W_{a, m}} \right) \quad (16)$$

where the coolant-flow ratio $W_c/W_{a,m}$ is a design parameter. The combustor pressure ratio $r_{3,2}$ is an input parameter.

Turbine. - The turbine work is equal to the fan work plus the compressor work:

$$(W_{a,m} - W_c + W_{f,m})\Delta h_T = W_{\text{tot}} \Delta h_F + W_{a,m} \Delta h_C \quad (17)$$

Substituting equation (16) into equation (17) and solving for turbine enthalpy drop yield

$$\Delta h_T = \frac{(1+b)\Delta h_F + \Delta h_C}{\left(1 - \frac{W_c}{W_{a,m}}\right)\left(1 + \frac{W_{f,m}}{W_{a,m} - W_c}\right)} \quad (18a)$$

Since

$$\Delta h\left(T'_3, T'_4, \frac{W_{f,m}}{W_{a,m} - W_c}\right) = \Delta h_T \quad (18b)$$

the turbine exit temperature T'_4 is determined from equation (18b). To determine the turbine pressure ratio, the ideal turbine-exit temperature $T'_{4,id}$ is needed. Therefore, the ideal turbine work is

$$\Delta h_{T,id} = \frac{\Delta h_T}{\eta_T} \quad (19a)$$

and

$$\Delta h\left(T'_3, T'_{4,id}, \frac{W_{f,m}}{W_{a,m} - W_c}\right) = \Delta h_{T,id} \quad (19b)$$

where turbine efficiency is a design parameter. The ideal turbine-exit temperature $T'_{4,id}$ is determined from equation (19b). Thus, the turbine pressure ratio is

$$\frac{P'_3}{P'_4} = \exp\left[\Delta\phi\left(T'_3, T'_{4,id}, \frac{W_{f,m}}{W_{a,m} - W_c}\right)\overline{M}_g \frac{J}{R}\right] \quad (20)$$

where the equation for the molecular weight of the gas is given in appendix B.

Coolant mixing. - The turbine-exit gas and coolant-flow mix just downstream of the turbine. This mixing is assumed to take place without any change in the mainstream total pressure. However, there is a change in total temperature. An energy balance for the mixing section with 0°R as the reference temperature yields

$$W_c \Delta h(T'_2, 0) + (W_{a,m} - W_c + W_{f,m}) \Delta h\left(T'_4, 0, \frac{W_{f,m}}{W_{a,m} - W_c}\right) \\ = (W_{a,m} + W_{f,m}) \Delta h\left(T'_{4M}, 0, \frac{W_{f,m}}{W_{a,m}}\right)$$

Dividing by $W_{a,m}$ yields

$$\frac{W_c}{W_{a,m}} \Delta h(T'_2, 0) + \left(1 - \frac{W_c}{W_{a,m}}\right) \left(1 + \frac{W_{f,m}}{W_{a,m} - W_c}\right) \Delta h\left(T'_4, 0, \frac{W_{f,m}}{W_{a,m} - W_c}\right) \\ = \left(1 + \frac{W_{f,m}}{W_{a,m}}\right) \Delta h\left(T'_{4M}, 0, \frac{W_{f,m}}{W_{a,m}}\right)$$

Solving for the exit enthalpy yields

$$\Delta h\left(T'_{4M}, 0, \frac{W_{f,m}}{W_{a,m}}\right) = \frac{\frac{W_c}{W_{a,m}} \Delta h(T'_2, 0) + \left(1 - \frac{W_c}{W_{a,m}}\right) \left(1 + \frac{W_{f,m}}{W_{a,m} - W_c}\right) \Delta h\left(T'_4, 0, \frac{W_{f,m}}{W_{a,m} - W_c}\right)}{1 + \frac{W_{f,m}}{W_{a,m}}} \quad (21)$$

Total exit temperature T'_{4M} is determined from equation (21).

The total pressure is

$$P'_{4M} = P'_4 \quad (22)$$

Afterburner. - Three cases are considered for the afterburner. For case I with no afterburner,

$$P'_5 = P'_4 \quad T'_5 = T'_{4M} \quad \frac{W_{f,AB}}{W_{a,m}} = 0 \quad (23)$$

For case II with the afterburner not lighted,

$$\frac{P'_5}{P'_4} = r_{5,4,n} \quad T'_5 = T'_{4M} \quad \frac{W_{f,AB}}{W_{a,m}} = 0 \quad (24)$$

where $r_{5,4,n}$ is given. For case III with afterburning, an energy balance on the afterburner yields

$$\eta_{AB} \frac{W_{f,AB}}{W_{a,m}} \text{HVF} + \left(1 + \frac{W_{f,m}}{W_{a,m}}\right) h'_{4M} = \left(1 + \frac{W_{f,m}}{W_{a,m}} + \frac{W_{f,AB}}{W_{a,m}}\right) h'_5 \quad (25)$$

Solving equation (25) for the afterburner fuel to air ratio yields

$$\frac{W_{f,AB}}{W_{a,m}} = \frac{\Delta h(T'_5, T_R) + \Delta h_b(T'_5, T_R) \frac{W_{f,m}}{W_{a,m}} - \left(1 + \frac{W_{f,m}}{W_{a,m}}\right) \Delta h(T'_{4M}, T_R, \frac{W_{f,m}}{W_{a,m}})}{\eta_{AB} \text{HVF} - \Delta h_b(T'_5, T_R)} \quad (26)$$

where the afterburner temperature T'_5 is a design parameter. The total-pressure ratio across the afterburner is

$$\frac{P'_5}{P'_4} = r_{5,4} \quad (27)$$

which is a design parameter. Therefore, the total fuel to air ratio of the mainstream is

$$\frac{W_{f,tot}}{W_{a,m}} = \frac{W_{f,AB}}{W_{a,m}} + \frac{W_{f,m}}{W_{a,m}} \quad (28)$$

Main nozzle. - Full expansion is assumed in the mainstream nozzle. Therefore,

$$\frac{P_6}{P'_5} = \frac{P_0}{P'_5} \quad (29)$$

and

$$\Delta\phi_N = \frac{R}{JM_g} \ln \frac{P_6}{P'_5} \quad (30a)$$

Since

$$\Delta\phi \left(T_{6, id}, T'_5, \frac{W_{f, tot}}{W_{a, m}} \right) = \Delta\phi_N \quad (30b)$$

the ideal nozzle exit temperature $T_{6, id}$ can be determined from equation (30b). The main nozzle-exit velocity is

$$V_6 = \psi \sqrt{2gJ \Delta h \left(T'_5, T_{6, id}, \frac{W_{f, tot}}{W_{a, m}} \right)} \quad (31)$$

where ψ is the velocity coefficient and a function of the airplane Mach number (ref. 1).

The equations derived thus far are for the main flow. The equations for the duct flow are now derived.

Duct burner. - Three cases are considered for the duct burner. For case I with no duct burner,

$$P'_7 = P'_1, \quad T'_7 = T'_1, \quad \frac{W_{f, D}}{W_{a, D}} = 0 \quad (32)$$

For case II with the duct burner not lighted,

$$\frac{P'_7}{P'_1} = r_{7, 1', n}, \quad T'_7 = T'_1, \quad \frac{W_{f, D}}{W_{a, D}} = 0 \quad (33)$$

For case III with duct burning, an energy balance on the duct burner is

$$W_{a, D} \Delta h(T'_{1'}, T_R) + W_{f, DB} (\eta_{DB})(HVF) = (W_{a, D} + W_{f, DB}) \Delta h_g \quad (34)$$

Dividing equation (34) by $W_{a,D}$ and solving for the duct-burner fuel to air ratio yield

$$\frac{W_{f,DB}}{W_{a,D}} = \frac{\Delta h(T'_7, T_R) - \Delta h(T'_1, T_R)}{\eta_{DB}^{HVF} - \Delta h_b(T'_7, T_R)} \quad (35)$$

where the duct-burner temperature T'_7 is a design parameter. The duct-burner fuel to air ratio based on the main flow is

$$\frac{W_{f,DB}}{W_{a,m}} = \frac{W_{f,DB}}{W_{a,D}} \frac{W_{a,D}}{W_{a,m}} = \frac{W_{f,DB}}{W_{a,D}} b \quad (36)$$

The total pressure ratio is

$$\frac{P'_7}{P'_{1'}} = r_{7,1'} \quad (37)$$

which is a design parameter.

Duct nozzle. - Full expansion is also assumed in the duct nozzle. Therefore,

$$\frac{P_8}{P'_7} = \frac{P_0}{P'_7} \quad (38)$$

and

$$\Delta\phi_{D,N} = \frac{R}{JM_g} \ln \frac{P_8}{P'_7} \quad (39a)$$

Since

$$\Delta\phi\left(T_{8,id}, T'_7, \frac{W_{f,DB}}{W_{a,D}}\right) = \Delta\phi_{D,N} \quad (39b)$$

the ideal exit temperature $T_{8,id}$ is determined from equation (39b). The duct nozzle-exit velocity is

$$V_8 = \psi \sqrt{2gJ \Delta h\left(T'_7, T_{8,id}, \frac{W_{f,DB}}{W_{a,D}}\right)} \quad (40)$$

where ψ is a function of M_0 (ref. 1).

Specific thrust. - The specific thrust of the engine is defined as the net thrust divided by the total airflow:

$$ST = \frac{(W_8 V_8 + W_6 V_6 - W_{tot} V_0)}{g W_{tot}} \quad (41)$$

Substituting the weight-flow relations yields

$$ST = \frac{(W_{a,D} + W_{f,DB})V_8 + (W_{a,m} + W_{f,tot})V_6 - (1+b)W_{a,m}V_0}{g(1+b)W_{a,m}} \quad (42)$$

and rearranging equation (42) yields

$$ST = \frac{\left(b + \frac{W_{f,DB}}{W_{a,m}}\right)V_8 + \left(1 + \frac{W_{f,tot}}{W_{a,m}}\right)V_6 - (1+b)V_0}{g(1+b)} \quad (43)$$

Specific fuel consumption. - Specific fuel consumption is defined as the total fuel flow in pounds per hour divided by the net thrust in pounds:

$$SFC = \frac{(W_{f,tot} + W_{f,DB})3600g}{(W_{a,D} + W_{f,DB})V_8 + (W_{a,m} + W_{f,tot})V_6 - (1+b)W_{a,m}V_0} \quad (44)$$

Dividing equation (44) by $W_{a,m}$ yields

$$SFC = \frac{\left(\frac{W_{f,tot}}{W_{a,m}} + \frac{W_{f,DB}}{W_{a,m}}\right)3600g}{\left(b + \frac{W_{f,DB}}{W_{a,m}}\right)V_8 + \left(1 + \frac{W_{f,tot}}{W_{a,m}}\right)V_6 - (1+b)V_0} \quad (45)$$

Engine efficiency. - Another performance parameter that is often used is engine efficiency:

$$\eta_e = \frac{\text{Thrust power}}{\text{Heat added}} \quad (46a)$$

The thrust power is equal to the net thrust multiplied by the inlet velocity. The heat added is

$$W_f(\text{HVF})J \quad (46b)$$

Therefore,

$$\eta_e = \frac{3600V_0}{(\text{SFC})(\text{HVF})J} \quad (47)$$

The method presented for determining the performance of a general jet engine applies to several engines: dry turbojet, afterburning turbojet, dry turbofan, and duct-burning turbofan. The performance of any one of these engines is obtained by eliminating the appropriate components of the general engine from the calculation procedure.

The analysis of a dry turbojet is obtained by eliminating the duct equations, setting the fan pressure ratio (P'_1/P'_1) equal to 1, the bypass ratio (b) equal to 0, the fan efficiency (η_F) equal to 1.0, and taking case I for the afterburner section. The analyses of the afterburning turbojet are the same as the dry turbojet except that case II or III for the afterburner section is used. The analyses of the turbofan engines are obtained by eliminating the afterburner, that is, by setting $P'_5 = P'_{4M}$, and $T'_5 = T'_{4M}$.

DESCRIPTION OF INPUT

A description of the input for the FORTRAN 7094 computer program is given in this section. The input quantities are turbine-inlet temperature, coolant-flow ratio, overall pressure ratio, fan pressure ratio, turbine efficiency, compressor efficiency, combustor efficiency, afterburner efficiency, duct-burner efficiency, bypass ratio, afterburner temperature, duct-burner temperature, airplane Mach number, altitude, temperature at this altitude, pressure at this altitude, combustor pressure ratio, afterburner pressure ratio, duct-burner pressure ratio, velocity coefficient of nozzle, diffuser pressure ratio, and heating value of fuel. The input is arranged so that parametric variation of the design parameters can be run.

The input format with sample data is shown in table I.

The definitions of the terms used on the cards follow:

Card 1:

R32	combustor pressure ratio
R54	afterburner pressure ratio when afterburning
R54N	afterburner pressure ratio when afterburner is not lighted
ETAC1	compressor efficiency, first value or only value
ETAT1	turbine efficiency, first value or only value
HVF	heating value of fuel, Btu/lb
NT3	number of turbine-inlet temperatures (must be an integer)
NP2	number of overall pressure ratios (must be an integer)

Card 2:

WCWA1	coolant-flow ratio, first value or only value
EAB1	afterburner efficiency, first value or only value
EB1	combustor efficiency, first value or only value
TR	enthalpy reference temperature, $^{\circ}\text{R}$
NM	number of airplane Mach numbers, must be equal to number of altitudes

Card 3:

ED1	duct-burner efficiency, first value or only value
DED	increment of duct-burner efficiency
EDM	maximum value of duct-burner efficiency
R711	duct-burner pressure ratio when burning
R711N	duct-burner pressure ratio when not burning
NP11	number of fan pressure ratios, must be an integer
NB	number of bypass ratios, must be an integer

Card 4:

DETAC	increment of compressor efficiency
ETACM	maximum compressor efficiency
DETAT	increment of turbine efficiency
ETATM	maximum turbine efficiency
DWCWA	increment of coolant-flow ratio
WCWAM	maximum coolant-flow ratio
DEB	increment of combustor efficiency
EBM	maximum combustor efficiency
DEAB	increment of afterburner efficiency
EABM	maximum afterburner efficiency

Card 5:

ETAF1	fan efficiency, first value or only value
DETAF	increment of fan efficiency
ETAFM	maximum fan efficiency

Card 6:

AM	airplane Mach number, maximum of 50
----	-------------------------------------

Card 7:

VCI	nozzle-velocity coefficient, one for each Mach number
-----	---

Card 8:

R10A	diffuser pressure ratio, one for each Mach number
------	---

Card 9:

ALTI	altitude, ft, maximum of 50
------	-----------------------------

Card 10:

TOSA	temperature of air at given altitude, °R
------	--

Card 11:

POSA	pressure of air at given altitude, inches of mercury
------	--

Card 12:

P2P1	overall pressure ratio, maximum of 50
------	---------------------------------------

Card 13:

P11P1	fan-pressure ratio, maximum of 50
-------	-----------------------------------

Card 14:

BA	bypass ratio, maximum of 50
----	-----------------------------

Card 15:

T3	turbine-inlet temperature, °R (maximum of 50)
----	---

Card 16:

T71	duct-burner temperature, °R (first value or only value)
DT7	increment of duct-burner temperature, °R
T7M	maximum duct-burner temperature, °R
T51	afterburner temperature, °R (first value or only value)

DT5 increment of afterburner temperature, $^{\circ}\text{R}$
T5M maximum afterburner temperature, $^{\circ}\text{R}$
KODE determines engine type, must be an integer
LD index on input statements, must be an integer (If $\text{LD} = 1$, all 16 data cards are read for the next set. If $\text{LD} = 2$, the next set of data consists only of card 16.)
KT5 determines when afterburner is used, must be an integer
KT7 determines when duct burner is used, must be an integer

The values of KODE, KT5, T51, KT7, and T71 for the various engines are given in table II.

TABLE II. - VALUES OF KODE, KT5, T51, KT7, AND T71

Engine	KODE	KT5	T51	KT7	T71
Dry turbojet	0	0	0.0	0	0.0
Afterburning turbojet, afterburner not lighted	0	-1	0.0	0	0.0
Afterburning turbojet	0	2	T_{AB}	0	0.0
Dry turbofan	2	0	0.0	0	0.0
Duct-burning turbofan	2	0	0.0	2	T_{DB}
Duct-burning turbofan, duct burner not lighted	2	0	0.0	-1	0.0

DESCRIPTION OF OUTPUT

An example of the output obtained is given in table III. The first four lines are input data, except for T1, total temperature at fan inlet and P1, total pressure at fan inlet. The next line contains the titles for the following columns:

P2P1	overall pressure ratio, which is an input and primary parametric variable
WFWA	overall fuel to air ratio, (lb fuel)/(lb air)
P4P3	turbine total-pressure ratio
ST	specific thrust, (lb thrust)/(lb air)
SFC	specific fuel consumption, (lb fuel)/(lb thrust)(hr)
T11	fan-exit or compressor-inlet total temperature, °R
T2	compressor-exit total temperature, °R
T4	turbine-exit total temperature, °R
T5	afterburner temperature (If there is no afterburning, T5 is equal to T4M, the total temperature of the gases at the exit of the mixing station.), °R
T7	duct-burner temperature (If there is no duct burning, T7 = T11.), °R
P11	fan-exit total pressure, psia
P2	compressor-exit total pressure, psia
P3	combustor-exit total pressure, psia
P4	turbine-exit total pressure, psia
P7	duct nozzle-inlet total pressure, psia
V6	main nozzle-exit velocity, ft/sec
V8	duct nozzle-exit velocity, ft/sec
ETA-E	overall engine efficiency

TABLE III. - SAMPLE OUTPUT

DRY TURBOFAN ENGINE CYCLE PROGRAM

M	ALT	O.	TOS	ETAC	ETAT	ETAF	HVF	RLO	R32	R711	VC	VCD	WCWA	V0	T1	P1	PUS
518.7	0.880	0.900	0.900	0.900	0.900	18640.	1.000	0.950	1.000	0.960	0.960	0.960	0.	0.	518.7	14.7	14.695
ER	EAB	ED	PSP4														
0.980	0.	C.	1.000														
R	PL1P1	T3															
0.500	2.50	2600.0															
P2P1	WFWA	P4P3	ST	SFC	T11	T2	T4	T5	T7	P11	P2	P3	P4	P7	V6	V8	ETA-E
3.00	0.01974	0.6322	57.11	1.2441	689.9	731.2	2315.6	2315.6	689.9	36.7	44.1	41.9	26.5	36.7	2034.5	1327.7	0.
4.00	0.01907	0.5631	61.93	1.1086	689.9	800.5	2315.6	2315.6	689.9	36.7	58.8	55.8	31.4	36.7	2262.7	1327.7	0.
5.00	0.01852	0.5100	64.77	1.0291	689.9	858.0	2269.1	2269.1	689.9	36.7	73.5	69.8	35.6	36.7	2398.2	1327.7	0.
6.00	0.01804	0.4672	66.62	0.9748	689.9	907.3	2228.7	2228.7	689.9	36.7	88.2	83.8	39.1	36.7	2486.5	1327.7	0.
7.00	0.01761	0.4316	67.88	0.9342	689.9	950.9	2192.6	2192.6	689.9	36.7	102.9	97.7	42.2	36.7	2547.2	1327.7	0.
8.00	0.01723	0.4013	68.76	0.9021	689.9	990.0	2159.9	2159.9	689.9	36.7	117.6	111.7	44.8	36.7	2590.2	1327.7	0.
9.00	0.01688	0.3750	69.38	0.8759	689.9	1025.5	2129.9	2129.9	689.9	36.7	132.3	125.6	47.1	36.7	2621.0	1327.7	0.
10.00	0.01656	0.3519	69.83	0.8537	689.9	1058.2	2102.0	2102.0	689.9	36.7	147.0	139.6	49.1	36.7	2643.1	1327.7	0.
11.00	0.01626	0.3314	70.13	0.8345	689.9	1088.4	2076.0	2076.0	689.9	36.7	161.6	153.6	50.9	36.7	2658.8	1327.7	0.
12.00	0.01598	0.3130	70.34	0.8177	689.9	1116.7	2051.5	2051.5	689.9	36.7	176.3	167.5	52.4	36.7	2669.6	1327.7	0.
R	PL1P1	T3															
1.000	2.50	2600.0															
P2P1	WFWA	P4P3	ST	SFC	T11	T2	T4	T5	T7	P11	P2	P3	P4	P7	V6	V8	ETA-E
3.00	0.01480	0.5492	48.81	1.0919	689.9	731.2	2304.3	2304.3	689.9	36.7	44.1	41.9	23.0	36.7	1763.3	1327.7	0.
4.00	0.01430	0.4872	52.92	0.9731	689.9	800.5	2243.5	2243.5	689.9	36.7	58.8	55.8	27.2	36.7	2022.5	1327.7	0.
5.00	0.01389	0.4397	55.29	0.9042	689.9	858.0	2201.8	2201.8	689.9	36.7	73.5	69.8	30.7	36.7	2172.9	1327.7	0.
6.00	0.01353	0.4015	56.82	0.8572	689.9	907.3	2161.0	2161.0	689.9	36.7	88.2	83.8	33.6	36.7	2269.9	1327.7	0.
7.00	0.01321	0.3698	57.85	0.8221	689.9	950.9	2124.6	2124.6	689.9	36.7	102.9	97.7	36.1	36.7	2336.1	1327.7	0.
8.00	0.01292	0.3428	58.57	0.7943	689.9	990.0	2091.5	2091.5	689.9	36.7	117.6	111.7	38.3	36.7	2382.7	1327.7	0.
9.00	0.01266	0.3195	59.08	0.7714	689.9	1025.5	2061.2	2061.2	689.9	36.7	132.3	125.6	40.1	36.7	2416.0	1327.7	0.
10.00	0.01242	0.2991	59.44	0.7521	689.9	1058.2	2033.1	2033.1	689.9	36.7	147.0	139.6	41.7	36.7	2439.7	1327.7	0.
11.00	0.01219	0.2809	59.69	0.7354	689.9	1088.4	2005.8	2005.8	689.9	36.7	161.6	153.6	43.1	36.7	2456.5	1327.7	0.
12.00	0.01198	0.2647	59.86	0.7207	689.9	1116.7	1982.1	1982.1	689.9	36.7	176.3	167.5	44.3	36.7	2467.9	1327.7	0.
R	PL1P1	T3															
0.500	2.50	3000.0															
P2P1	WFWA	P4P3	ST	SFC	T11	T2	T4	T5	T7	P11	P2	P3	P4	P7	V6	V8	ETA-E
3.00	0.02488	0.6765	63.61	1.4084	689.9	731.2	2781.0	2781.0	689.9	36.7	44.1	41.9	28.3	36.7	2321.6	1327.7	0.
4.00	0.02420	0.6137	69.19	1.2593	689.9	800.5	2728.4	2728.4	689.9	36.7	58.8	55.8	34.3	36.7	2586.2	1327.7	0.
5.00	0.02364	0.5648	72.60	1.1720	689.9	858.0	2684.2	2684.2	689.9	36.7	73.5	69.8	39.4	36.7	2745.6	1327.7	0.
6.00	0.02315	0.5250	74.91	1.1125	689.9	907.3	2645.7	2645.7	689.9	36.7	88.2	83.8	44.0	36.7	2855.1	1327.7	0.
7.00	0.02271	0.4915	76.55	1.0682	689.9	950.9	2611.4	2611.4	689.9	36.7	102.9	97.7	48.0	36.7	2933.7	1327.7	0.
8.00	0.02232	0.4627	77.77	1.0333	689.9	990.0	2580.4	2580.4	689.9	36.7	117.6	111.7	51.7	36.7	2992.3	1327.7	0.
9.00	0.02197	0.4376	78.70	1.0049	689.9	1025.5	2551.9	2551.9	689.9	36.7	132.3	125.6	55.0	36.7	3037.1	1327.7	0.
10.00	0.02164	0.4153	79.41	0.9809	689.9	1058.2	2525.5	2525.5	689.9	36.7	147.0	139.6	58.0	36.7	3071.9	1327.7	0.
11.00	0.02133	0.3953	79.96	0.9603	689.9	1088.4	2500.8	2500.8	689.9	36.7	161.6	153.6	60.7	36.7	3099.3	1327.7	0.
12.00	0.02104	0.3773	80.40	0.9423	689.9	1116.7	2477.6	2477.6	689.9	36.7	176.3	167.5	63.2	36.7	3120.9	1327.7	0.
R	PL1P1	T3															
1.000	2.50	3000.0															
P2P1	WFWA	P4P3	ST	SFC	T11	T2	T4	T5	T7	P11	P2	P3	P4	P7	V6	V8	ETA-E
3.00	0.01866	0.6009	54.29	1.2374	689.9	731.2	2717.7	2717.7	689.9	36.7	44.1	41.9	25.2	36.7	2040.8	1327.7	0.
4.00	0.01815	0.5436	58.90	1.1095	689.9	800.5	2664.7	2664.7	689.9	36.7	58.8	55.8	30.4	36.7	2379.2	1327.7	0.
5.00	0.01773	0.4991	61.68	1.0348	689.9	858.0	2620.2	2620.2	689.9	36.7	73.5	69.8	34.8	36.7	2553.7	1327.7	0.
6.00	0.01736	0.4629	63.53	0.9837	689.9	907.3	2581.4	2581.4	689.9	36.7	88.2	83.8	38.8	36.7	2671.0	1327.7	0.
7.00	0.01704	0.4325	64.85	0.9457	689.9	950.9	2546.9	2546.9	689.9	36.7	102.9	97.7	42.3	36.7	2754.9	1327.7	0.
8.00	0.01674	0.4065	65.83	0.9156	689.9	990.0	2515.6	2515.6	689.9	36.7	117.6	111.7	45.4	36.7	2817.1	1327.7	0.
9.00	0.01647	0.3837	66.56	0.8910	689.9	1025.5	2486.9	2486.9	689.9	36.7	132.3	125.6	48.2	36.7	2864.6	1327.7	0.
10.00	0.01623	0.3635	67.13	0.8702	689.9	1058.2	2460.3	2460.3	689.9	36.7	147.0	139.6	50.8	36.7	2901.5	1327.7	0.
11.00	0.01600	0.3455	67.57	0.8523	689.9	1088.4	2435.4	2435.4	689.9	36.7	161.6	153.6	53.1	36.7	2930.4	1327.7	0.
12.00	0.01578	0.3293	67.92	0.8366	689.9	1116.7	2412.0	2412.0	689.9	36.7	176.3	167.5	55.2	36.7	2953.2	1327.7	0.

PROGRAM VARIABLES

ALT	altitude
AM1	airplane Mach number
B	bypass ratio
CP(T, F)	specific heat, function of temperature and fuel to air ratio
CP1	temporary storage of CP(T, F)
DDH	difference between Δh and $\Delta h(T_2, T_1, f)$
DDS	difference between $\Delta \phi$ and $\Delta \phi(T_2, T_1, f)$
DDSC	difference between $\Delta \phi$ and $\Delta \phi(T_2, T_1, f)$ for compressor
DHC	enthalpy change across compressor
DHF	enthalpy change across fan
DHT	enthalpy change across turbine
DH4M	enthalpy at station 4M
DSC	eq. (11)
DSF	eq. (7)
DSN	eq. (30)
EAB	afterburner efficiency
EB	combustor efficiency
ED	duct-burner efficiency
ETAC	compressor efficiency
ETAF	fan efficiency
ETAT	turbine efficiency
GAMMA	ratio of specific heats
H(T ₂ , T ₁ , F)	enthalpy change of gas
HA(T ₂ , T ₁)	enthalpy change of air
HF(T ₂ , T ₁)	Δh_p of eqs. (14) and (B8c)
POPOS	total- to static-pressure ratio at diffuser inlet
POS	pressure of air at flight altitude
P2P11	compressor pressure ratio

P5P4	pressure ratio across afterburner
P7P11	pressure ratio across duct burner
R10	temporary storage of R10A
$S(T_2, T_1, F)$	entropy function
TOS	temperature of air at flight altitude
TO	total temperature at diffuser inlet
TO1	temporary storage of TO
T111	another name for T11
T11D	temporary storage of T11
T11ID	ideal fan-exit total temperature
T21	temporary storage of T2
T21D	ideal total temperature of compressor exit
T21D1	temporary storage of T21D
T31	temporary storage of T3
T41	temporary storage of T4
T41D	ideal turbine-exit total temperature
T41D1	temporary storage of T41D
T4M	total temperature at station 4M
T4M1	temporary storage of T4M
T5	nozzle-inlet total temperature
T6S	static temperature at main nozzle exit
T6S1	temporary storage of T6S
T7	duct nozzle-inlet total temperature
T8S	static temperature at duct nozzle exit
T8S1	temporary storage of T8S
VO	airplane velocity
VC	nozzle-velocity coefficient
W(F)	molecular weight, function of fuel to air ratio
WCWA	coolant flow ratio

WFAWA fuel to air ratio of afterburner
WFDWA fuel to air ratio of duct burner
WFMWA fuel to air ratio of main combustor, based on compressor airflow
WFWA1 fuel to air ratio in combustor
WFWA total fuel to air ratio

PROGRAM LISTING

```

C   JET ENGINE CYCLE PROGRAM
C   DRY TURBOJET  KODE = 0 T5= 0.0 KT5 = 0
C   TURBOJET AFTERBURNER NOT LIT KODE=0 T5=0.0 KT5=-1
C   AFTERBURNING TURBOJET KODE = 0 T5 = AFTERBURNER TEMP KT5 = 2
C   DRY TURBOFAN KODE = 2 T7= 0.0 KT5 = 0 KT7 = 0
C   DUCT-BURNING TURBOFAN KODE = 2 T7 = D.B. TEMP KT7 = 2 KT5 = 0
C   TURBOFAN DUCT BURNER NOT LIT KODE=2 T7=0.0 KT7=-1 KT5=0
C
C   DIMENSION T3(53), P2P1(50), T5A(50), AM(50), VC1(53), R10A(50),
1TOSA(50), POSA(50), ALTI(50), P11P1(50), BA(50)
C   THERMODYNAMIC PROPERTIES
CP(T,F)=(1.0/(1.0+F))*(AC-BB*T+CC*T**2-DD*T**3
1+EE*T**4+(AAA+BBB*T-CCC*T**2+DDD*T**3
2-EEE*T**4)*F)
W(F)=(1.0+F)/( .034522+.035648*F)
H(T2,T1,F)=(1.0/(1.0+F))*(AC*(T2-T1)-(BB/2.0)*(T2**2
1-T1**2)+(CC/3.0)*(T2**3-T1**3)-(DD/4.0)*(T2**4-
2T1**4)+(EE/5.0)*(T2**5-T1**5)+(AAA*(T2-T1)
3+(BBB/2.)*(T2**2-T1**2)-(CCC/3.0)*(T2**3-T1**3)
4+(DDD/4.0)*(T2**4-T1**4)-(EEE/5.0)*(T2**5-T1**5))
5*F)
S(T2,T1,F)=(1.0/(1.0+F))*(AC*ALOG(T2/T1)-BB*
1(T2-T1)+(CC/2.0)*(T2**2-T1**2)-(DD/3.0)*(T2**3-T1**3)
2+(EE/4.0)*(T2**4-T1**4)+F*(AAA*ALOG(T2/T1)+
3BBB*(T2-T1)-(CCC/2.0)*(T2**2-T1**2)+(DDD/3.0)*
4(T2**3-T1**3)-(EEE/4.0)*(T2**4-T1**4)))
HA(T2,T1)=AC*(T2-T1)-(BB/2.0)*(T2**2-T1**2)
1+(CC/3.0)*(T2**3-T1**3)-(DD/4.0)*(T2**4-T1**4)
2+(EE/5.0)*(T2**5-T1**5)
HF(T2,T1)=(AAA*(T2-T1)+(BBB/2.0)*(T2**2-T1**2)
1-(CCC/3.0)*(T2**3-T1**3)+(DDD/4.0)*(T2**4-T1**4)
2-(EEE/5.0)*(T2**5-T1**5))
AC=.24062
BB=.017724E-3
CC=.038506E-6
DD=.012662E-9
EE=.0013012E-12
AAA=.22091
BBB=.51822E-3
CCC=.19462E-6
DDD=.045089E-9
EEE=.0043275E-12
10 READ(5,50)R32,R54,R54N,ETAC1,ETAT1,HVF,NT3,NP2,NT5
50 FORMAT(6F7.3,3I3)
READ(5,54)WCWA1,EAB1,EB1,TR,NM
54 FORMAT(4F6.1,I3)
READ(5,56)ED1,DED,EDM,R711,R711N,NP11,NB
56 FORMAT(5F6.1,2I3)
READ(5,55) DETAC,ETACM,DETAT,ETATM,DWCWA,WCWAM,DEB,EBM,DEAB,EABM
55 FORMAT(10F6.1)
READ(5,53)ETAF1,DETAf,ETAfM

```

```

      READ(5,53)(AM(I),I=1,NM)
      READ(5,53)(VC1(I),I = 1,NM)
      READ(5,53)(R10A(I),I = 1,NM)
      READ(5,53)(ALTI(I),I=1,NM)
      READ(5,53)(TOSA(I),I = 1,NM)
      READ(5,53)(POSA(I),I = 1,NM)
53  FORMAT(12F6.3)
      READ(5,52)(P2P1(J), J = 1,NP2)
52  FORMAT(12F6.2)
      READ(5,52)(P11P1(J) ,J=1,NP11)
      READ(5,51)(BA(K),K = 1,NB)
      READ(5,51)(T3(J),J=1,NT3)
51  FORMAT(12F6.1)
9   READ(5,57)T71,DT7,T7M,T51,DT5,T5M,KODE,LD,KT5,KT7
57  FORMAT(6F6.1,4I3)
      R=53.3
      CJ=778.0
      G=32.2
      DO 103 I = 1,NM
      TOS = TOSA(I)
      POS = POSA(I)
      ALT = ALTI(I)
      AM1=AM(I)
      VC = VC1(I)
      R10 = R10A(I)
      ETAC = ETAC1
      ETAT = ETAT1
      EB = EB1
      WCWA = WCWA1
      EAB = EAB1
      ED = ED1
      ETAF = ETAF1
      T5 = T51
      T7 = T71
      VCD = VC
      POS = POS* .49115
400 IF ( KODE .EQ. 0) GO TO 500
      IF (KT7) 64,65,66
65  WRITE(6,303)
303 FORMAT(1H1,20X,33H DRY TURBOFAN ENGINE CYCLE PROGRAM)
      P7P11 = 1.0
      GO TO 80
64  WRITE(6,304)
304 FORMAT(1H1,25X,29H TURBOFAN ENGINE CYCLE PROGRAM)
      P7P11 = R711N
      GO TO 80
66  WRITE(6,305)
305 FORMAT(1H1,20X,41H DUCT BURNING TURBOFAN ENGINE CYCLE PROGRAM)
      P7P11 = R711
      GO TO 80
500 WRITE(6,200)
200 FORMAT(1H1,30X,29H TURBOJET ENGINE CYCLE PROGRAM)
      IF (KT5) 63,62,61
62  WRITE(6,300)
300 FORMAT(1H0,14H NO AFTERBURNER,5X,11HP5/P4 = 1.0)
      GO TO 80

```

```

61 WRITE(6,301) R54
301 FORMAT(1H0,17HWITH AFTERBURNING,5X,7HP5/P4 =,F5.4)
GO TO 80
63 WRITE(6,302) R54N
302 FORMAT(1H0,35HNO AFTERBURNING WITH AN AFTERBURNER,
15X,7HP5/P4 =,F5.4)
80 CONTINUE
IF(KT5.EQ. 0) P5P4 = 1.0
IF(KT5.EQ.(-1 )) P5P4 = R54N
IF (KT5.GT. 0) P5P4 = R54
WRITE(6,206)
206 FORMAT(1H0,3X,1HM,10X,3HALT)
WRITE(6,207)AM1,ALT
207 FORMAT(1H,F6.3,F12.1)
IF (AM1 .EQ. 0.0) GO TO 30
GAMMA=1.0/(1.0-R/(CP(TOS,0.0)*CJ))
VO = AM1*SQRT(G*R*GAMMA*TOS)
CP1=CP(TOS,0.0)
T01 = 0.0
2 T0=TOS+VO**2/(2.0*G*CJ*CP1)
CP1=H(T0,TOS,0.0)/(T0-TOS )
IF (ABS(T0-T01) .LT. .5) GO TO 5
T01 = T0
GO TO 2
5 POPOS=EXP(S(T0,TOS,0.0)*CJ/R)
GO TO 40
30 T0 =TOS
VO = 0.0
POPOS = 1.0
C DIFFUSER
40 T1 = T0
PO = POPOS*POS
P1 = R10*PO
WRITE(6,1010)
1010 FORMAT( 117HJ TOS ETAC ETAT ETAF HVF
1 R10 R32 R711 VC VCD WCWA VO T1 P1
2 POS )
WRITE(6,1020) TOS,ETAC,ETAT,ETAF,HVF,R10,R32,R711,VC,VCD,WCWA,
1VO,T1,P1,POS
1020 FORMAT(1H , F7.1,3(F7.3),F9.0,5(F7.3),F8.4,2(F8.1),F8.1,
1F7.3)
WRITE(6,199)
199 FORMAT(1H0,2X,2HEB,5X,3HEAB,4X,2HED,4X,4HP5P4)
WRITE(6,201)EB,EAB,ED,P5P4
201 FORMAT(1H , F6.3,3F7.3)
81 DO 101 K = 1,NT3
T31=T3(K)
DO 102 L = 1,NP11
C FAN
DSF = (R/CJ)*ALOG(P11P1(L))
T11ID1 = 0.0
T11ID = T1*EXP(DSF/.24062)
310 DDS = DSF-S(T11ID,T1,0.0)
T11ID = T11ID + DDS*T11ID/CP(T11ID,0.0)
IF(ABS(T11ID-T11ID1).LT..5) GO TO 311
T11ID1 = T11ID

```

```

      GO TO 310
311 DHF = H(T11D,T1,0.0)/ETAF
      T11D = 0.0
      T11 = DHF/.24062 + T1
312 DDH = DHF - H(T11,T1,0.0)
      T11 = T11 + DDH/CP(T11,0.0)
      IF (ABS(T11-T11D).LT..5) GO TO 314
      T11D = T11
      GO TO 312
314 T111 = T11
      DO 104 KB = 1,NB
      B = BA(KB)
      WRITE(6,1030)
1030 FORMAT( 36HJ      B      P11P1      T3      )
      WRITE(6,1040) B,P11P1(L),T3(K)
1040 FORMAT(2XF6.3,F9.2,F9.1)
      WRITE(6,1050)
1050 FORMAT( 132H P2P1      WFWA      P4P3      ST      SFC      T11      T2      T
      14      T5      T7      P11      P2      P3      P4      P7      V6      V8
      2  ETA-E      )
      DO 100 J = 1,NP2
C   COMPRESSOR
      T21 = 0.0
      T21D1 = 0.0
      P2P11 = P2P1(J)/P11P1(L)
      DSC = (R/CJ)*ALOG(P2P11)
      T21D = T11*EXP(DSC/.24062)
18 DDSC = DSC-S(T21D,T11,0.0)
      T21D=T21D+DDSC*T21D/CP(T21D,0.0)
      IF(ABS(T21D-T21D1).LT..5) GO TO 6
      T21D1=T21D
      GO TO 18
6 DHC= H(T21D,T11,0.0)/ ETAC
      T2=DHC/.24062+T11
110 DDH=DHC-H(T2 ,T11,0.0)
      T2=T2+DDH/CP(T2,0.0)
      IF(ABS(T2-T21).LT. .5) GO TO 7
      T21=T2
      GO TO 110
C   COMBUSTOR
7 WFWA1=(HA(T31,TR)-HA(T2,TR))/(HVF*EB
1-HF(T31,TR))
      WFWA=WFWA1*(1.0-WCWA)
C   TURBINE
      DHT =(DHC+(1.+B)*DHF)/((1.+WFWA1)*(1.-WCWA))
      T4=T3(K)-DHT/.24062
      T41 = 0.0
15 DDH = H(T31,T4,WFWA1) - DHT
      T4=T4+DDH/CP(T4,WFWA1)
      IF(ABS(T4-T41).LT. .5) GO TO 16
      T41=T4
      GO TO 15
16 DHT1D=DHT/ETAT
      T41D=T3(K)-DHT1D/.24062
      T41D1 = 0.0
112 DDH = H(T31,T41D,WFWA1) - DHT1D

```

```

      T41D=T41D+DDH/CP(T41D,WFWA1)
      IF(ABS(T41D-T41D1).LT. .5) GO TO 111
      T41D1=T41D
      GO TO 112
111  IF ( T41D)1007,1007,1111
1111 P4P3=EXP(S(T41D,T31,WFWA1)*W(WFWA1)*.50346)
      IF (WCWA .EQ. 0.0) GO TO 24
C    COOLANT MIXING
      DH4M=(WCWA*H(T2,0.0,0.0)+(1.-WCWA)*(1.+WFWA1)*
      1H(T4,0.0,WFWA1))/(1.+WFMWA)
381  T4M=DH4M/.24062
      T4M1 = 0.0
25   DDH=DH4M-H(T4M,0.0,WFMWA)
      T4M=T4M+DDH/CP(T4M,WFMWA)
      IF(ABS(T4M-T4M1).LT. .5) GO TO 26
      T4M1=T4M
      GO TO 25
24   T4M = T4
26   IF(KT5.NE.0) GO TO 17
C    AFTERBURNER
      T5 = T4M
      P5P4 = 1.0
      WFWA = WFMWA
      WAWF=1.0/WFMWA
      GO TO 19
17   CONTINUE
      IF ( KT5.NE.(-1)) GO TO 20
      T5 = T4M
      P5P4 = R54N
      WAWF=1.0/WFMWA
      WFWA = WFMWA
      GO TO 19
20   CONTINUE
      IF (T5 .LT. T4M) GO TO 99
      P5P4 = R54
      WFAWA=(HA(T5,TR)+HF(T5,TR)*WFMWA-(1.+WFMWA)
      1*H(T4M,TR,WFMWA))/(HVF*EAB-HF(T5,TR))
      WFWA=WFAWA+WFMWA
      WAWF=1.0/WFWA
19   P11 = P1*P11P1(L)
      P2 = P2P11*P11
      P3 = P2*R32
      P4 = P4P3*P3
      P5 = P5P4*P4
C    MAIN NOZZLE
      POSP5 = POS/P5
      P6S = POS
      IF ( P5 .LT. P6S ) GO TO 1005
      IF(WFWA.GT. .06763) GO TO 105
      DSN=1.9862*ALOG(P6S/P5)/W(WFWA)
      T6S=T5*EXP(DSN/.24062)
      T6S1 = 0.0
11   DDS = DSN-S(T6S,T5,WFWA)
      T6S=T6S+DDS*T6S/CP(T6S,WFWA)
      IF (ABS(T6S-T6S1) .LT..5) GO TO 12
      T6S1 = T6S

```

```

      GO TO 11
* 12 V6=VC*SQRT(2.0*G*CJ*H(T5,T6S,WFWA))
      IF ( KODF.EQ. 0) GO TO 414
C   DUCT-BURNER
319 IF ( KT7.NE. 0) GO TO 320
      T7 = T111
      P7P11 = 1.0
      WFDWA = 0.0
      GO TO 350
320 IF ( KT7.NE.(-1)) GO TO 321
      T7 = T111
      P7P11 = R711N
      WFDWA = 0.0
      GO TO 350
321 IF ( T7.LT.T11) GO TO 98
      P7P11 = R711
      WFDWA = (HA(T7,TR)-HA(T111,TR))/(HVF*ED-HF(T7,TR))
C   DUCT NOZZLE
350 P7 = P7P11*P11
      DSN = 1.9862*ALOG(PDS/P7)/W(WFDWA)
      T8S = T7*EXP(DSN/.24062)
      T8S1 = 0.0
411 DDS = DSN-S(T8S,T7,WFDWA)
      T8S=T8S+DDS*T8S/CP(T8S,WFDWA)
      IF (ABS(T8S-T8S1) .LT..5) GO TO 412
      T8S1 = T8S
      GO TO 411
412 V8 = VCD*SQRT(2.0*G*CJ *H(T7,T8S,WFDWA))
      WFDWA = WFDWA*B
      ST = ((1.+WFMWA)*V6+(B+WFDWA)*V8-(1.+B)*V0)/(G*(1.+B))
      WFWA = WFDWA + WFMWA
      SFC = 3600.0*G*WFWA/((1.+WFMWA)*V6+(B+WFDWA)*V8-(1.+B)*V0)
      WFWA = WFWA/(1.+B)
      GO TO 415
414 ST = ((1.0+1.0/WAWF)*V6-V0)/G
      SFC = 3600.0*G*(1.0/WAWF)/((1.0+1.0/WAWF)*V6-V0)
415 ETAE = 3600.0*V0/(SFC*HVF*CJ)
      IF ( KODE .NE. 0 ) GO TO 416
      V8 = 0.0
      P7 = 0.0
      T7 = 0.0
416 WRITE(6,1060) P2P1(J),WFWA,P4P3,ST,SFC,T11,T2,T4, T5, T7,P11,P2,
      1P3,P4,P7,V6,V8,ETAE
1060 FORMAT(1X,F5.2,F8.5,F7.4,F7.2,F8.4,5F7.1,5F7.1,2F7.1,F6.3)
      GO TO 100
      99 WRITE(6,210)
210 FORMAT(1H0,19HT5 IS LESS THAN T4M)
      GO TO 100
      98 WRITE(6,211)
211 FORMAT(1H0,19HT7 IS LESS THAN T11)
      GO TO 100
105 WRITE(6,106) T3(K), P2P1(J),WFWA
      106 FORMAT(1H0,4HT3 =,F8.1,3X,7HP2P1 = ,F4.1,4X,7HWF/WA =,F8.6)
      GO TO 100
1005 WRITE(6,1006)
1006 FORMAT(1H0,21H P5 IS LESS THAN P6S )

```

```

      GO TO 100
1007 WRITE(6,1008)
1008 FORMAT(1H0,18H T41D IS NEGATIVE )
100  CONTINUE
    104 CONTINUE
    102 CONTINUE
    101 CONTINUE
      T5 = T5 + DT5
      IF (T5.LT.T5M) GO TO 81
      T5 = T51
      T7 = T7 + DT7
      IF (T7.LT. T7M) GO TO 81
      IF (ED.GE.EDM) GO TO 405
      ED = ED + DED
      GO TO 400
405  IF ( WCWA .GE. WCWAM ) GO TO 401
      WCWA = WCWA + DWCWA
      GO TO 400
401  IF ( EB .GE. EBM) GO TO 402
      EB = EB + DER
      GO TO 400
402  IF ( EAB .GE. EABM) GO TO 403
      EAB = EAB + DEAB
      GO TO 400
403  IF (ETAT .GE. ETATM) GO TO 404
      ETAT = ETAT +DETAT
      GO TO 400
404  IF (ETAC.GE. ETACM) GO TO 406
      ETAC = ETAC + DETAC
      GO TO 400
406  IF(ETAF.GE.ETAFM) GO TO 103
      ETAF = ETAF+DETAF
      GO TO 400
103  CONTINUE
      GO TO (10,9),LD
      END

```

Lewis Research Center,
 National Aeronautics and Space Administration,
 Cleveland, Ohio, November 22, 1966,
 720-03-01-35-22.

APPENDIX A

SYMBOLS

b	bypass ratio	γ	ratio of specific heats
C_p	specific heat, Btu/(lb)($^{\circ}$ R)	η	efficiency
f	fuel to air ratio, (lb fuel)/(lb air)	$\Delta\phi$	entropy function
g	gravitational constant, 32.17 ft/sec ²	ψ	velocity coefficient
HVF	heating value of fuel, Btu/lb	Subscripts:	
h	enthalpy, Btu/lb	AB	afterburner
Δh	enthalpy change, Btu/lb	a	air
Δh_b	correction to air enthalpy (eq. (B8c)), Btu/lb	B	combustor
J	mechanical equivalent of heat, 778 ft-lb/Btu	C	compressor
M	Mach number	c	coolant
\bar{M}	molecular weight	D	duct
P	pressure, psia	DB	duct burner
R	gas constant, ft-lb/(lb)($^{\circ}$ R)	e	engine
\mathcal{R}	universal gas constant, ft-lb/(lb mole)($^{\circ}$ R)	F	fan
r	pressure ratio	f	fuel
SFC	specific fuel consumption, (lb fuel)/(lb thrust)(hr)	g	gas
ST	specific thrust, (lb thrust)/(lb air)	id	ideal
T	temperature, $^{\circ}$ R	m	main
V	velocity, ft/sec	N	nozzle
W	weight flow, lb/hr	n	not lighted
\bar{W}	weight of component per pound of air, lb/lb air	R	reference
		T	turbine
		t	thrust
		tot	total
		0	diffuser inlet

1 fan inlet
1' fan outlet
2 compressor outlet, main burner
inlet
3 main burner outlet, turbine inlet
4 turbine outlet
4M mixing station

5 afterburner outlet
6 nozzle outlet
7 duct-burner outlet
8 duct nozzle outlet

Superscript:


(') total, as applied to state points

APPENDIX B


DERIVATION OF COMBUSTION GAS THERMODYNAMIC-PROPERTY EQUATIONS

Reaction Stoichiometry

The fuel used was of the form $C_n H_{2n}$. Therefore, the general combustion equation is



Eliminating n , the reaction and the formula weights are



For f pounds of fuel used, the amount of O_2 used is

$$\frac{48.000}{14.026} f = 3.422 f$$

The amount of CO_2 formed is

$$\frac{44.010}{14.026} f = 3.138 f$$

and the amount of H_2O formed is

$$\frac{18.016}{14.026} f = 1.284 f$$

The weights of the components of air per pound of air are oxygen, 0.2314; nitrogen, 0.7552; argon, 0.0129; and carbon dioxide, 0.0005. Therefore, the net amounts of components left after the reaction of f pounds of fuel with 1 pound of air are

$$\bar{W}_{O_2} = 0.2314 - 3.422 f$$

$$\bar{W}_{\text{CO}_2} = 0.0005 + 3.138 f$$

$$\bar{W}_{\text{H}_2\text{O}} = 1.284 f$$

$$\bar{W}_{\text{N}_2} = 0.7552$$

$$\bar{W}_{\text{Ar}} = 0.0129$$

and the weight of the gas is $1 + f$ pounds per pound of air.

Specific Heat

The specific heat of the gas is

$$(C_p)_g = \frac{\sum (\bar{W} C_p)_{\text{components}}}{1 + f} \quad (\text{B3})$$

The specific heat of each component is expressed in the form (ref. 2)

$$C_p = A + B(T \times 10^{-3}) + C(T \times 10^{-3})^2 + D(T \times 10^{-3})^3 + E(T \times 10^{-3})^4 \quad (\text{B4})$$

where C_p is in Btu per pound mass per $^{\circ}\text{R}$ and T is in $^{\circ}\text{R}$. The coefficients A , B , C , D , and E , are obtained from reference 2. These coefficients and the molecular weight of each component are given in table IV.

TABLE IV. - COEFFICIENTS AND MOLECULAR WEIGHTS OF COMPONENTS

Components	Coefficients in eq. (B4)					Molecular weight, \bar{M}
	A	B	C	D	E	
Nitrogen	0.25410	-0.032690	0.047685	-0.014973	0.0014885	28.016
Oxygen	.20334	.029680	.0089971	-.0058842	.00076764	32.000
Carbon dioxide	.11097	.21110	-.088140	.018003	-.0014317	44.010
Water vapor	.44266	-.033155	.087761	-.024552	.0021734	18.016
Argon	.12438	0	0	0	0	39.944

Therefore, the equation for the specific heat of the combustion products is obtained by substituting equation (B4) into equation (B3). The resulting equation is

$$(C_p)_g = \left(\frac{1}{1+f} \right) \left\{ 0.24062 - 0.017724 T \times 10^{-3} + 0.038056 \times (10^{-3} T)^2 - 0.012662 \times (10^{-3} T)^3 \right. \\ \left. + 0.0013012 \times (10^{-3} T)^4 + \left[0.22091 + 0.51822 \times 10^{-3} T \right. \right. \\ \left. \left. - 0.19462 \times (10^{-3} T)^2 + 0.045089 \times (10^{-3} T)^3 - 0.0043275 \times (10^{-3} T)^4 \right] f \right\} \quad (B5)$$

$$C_p(T, f) = (C_p)_g$$

Enthalpy Change

The enthalpy change can be expressed as

$$\Delta h = \int_{T_1}^{T_2} C_p dT \quad (B6)$$

Substituting equation (B5) into equation (B6) and integrating yield

$$\Delta h_g = \frac{1}{1+f} \left\{ 0.24062(T_2 - T_1) - \frac{0.017724 \times 10^{-3}}{2} (T_2^2 - T_1^2) + \frac{0.038056 \times 10^{-6}}{3} (T_2^3 - T_1^3) \right. \\ \left. - \frac{0.012662 \times 10^{-9}}{4} (T_2^4 - T_1^4) + \frac{0.0013012 \times 10^{-12}}{5} (T_2^5 - T_1^5) + \left[0.22091(T_2 - T_1) \right. \right. \\ \left. \left. + \frac{0.51822 \times 10^{-3}}{2} (T_2^2 - T_1^2) - \frac{0.19462 \times 10^{-6}}{3} (T_2^3 - T_1^3) + \frac{0.045089 \times 10^{-9}}{4} (T_2^4 - T_1^4) \right. \right. \\ \left. \left. - \frac{0.0043275 \times 10^{-12}}{5} (T_2^5 - T_1^5) \right] f \right\} \quad (B7)$$

$$\Delta h(T_2, T_1, f) = \Delta h_g$$

The enthalpy change of the gas can also be expressed as

$$\Delta h_g = \frac{1}{1+f} (\Delta h_a + \Delta h_b f) \quad (B8a)$$

where Δh_a is the enthalpy change of 1 pound of air

$$\Delta h_a = \Delta h(T_2, T_1) \quad (B8b)$$

and Δh_b is the correction to the air enthalpy due to combustion and is expressed as

$$\begin{aligned} \Delta h_b = & 0.22091(T_2 - T_1) + \frac{0.51822 \times 10^{-3}}{2} (T_2^2 - T_1^2) - \frac{0.19462 \times 10^{-6}}{3} (T_2^3 - T_1^3) \\ & + \frac{0.045089 \times 10^{-9}}{4} (T_2^4 - T_1^4) - \frac{0.0043275 \times 10^{-12}}{5} (T_2^5 - T_1^5) \end{aligned} \quad (B8c)$$

Entropy Function

For isentropic flow,

$$\frac{R}{J} \ln \frac{P_2}{P_1} = \int_{T_1}^{T_2} \frac{C_p}{T} dT \quad (B9)$$

This equation is used in the evaluation of ideal processes in turbomachines. For convenience, the right side of equation (B9) will herein be called the entropy function. The entropy function is

$$\Delta \phi = \int_{T_1}^{T_2} \frac{C_p}{T} dT \quad (B10)$$

Substituting equation (B5) into equation (B10) and integrating yield

$$\begin{aligned}
\Delta \phi_g' = \frac{1}{1+f} \left\{ 0.24062 \ln \frac{T_2}{T_1} - 0.017724 \times 10^{-3} (T_2 - T_1) + \frac{0.038056 \times 10^{-6}}{2} (T_2^2 - T_1^2) \right. \\
- \frac{0.012662 \times 10^{-9}}{3} (T_2^3 - T_1^3) + \frac{0.0013012 \times 10^{-12}}{4} (T_2^4 - T_1^4) \\
+ f \left[0.22091 \ln \frac{T_2}{T_1} + 0.51822 \times 10^{-3} (T_2 - T_1) - \frac{0.19462 \times 10^{-6}}{2} (T_2^2 - T_1^2) \right. \\
\left. \left. + \frac{0.045089 \times 10^{-9}}{3} (T_2^3 - T_1^3) - \frac{0.0043275 \times 10^{-12}}{4} (T_2^4 - T_1^4) \right] \right\} \quad (B11)
\end{aligned}$$

$$\Delta \phi(T_2, T_1, f) = \Delta \phi_g$$

Molecular Weight

The molecular weight of the gas is equal to the weight of the gas divided by the total number of moles (sum of the moles of the components). Therefore, molecular weight can be expressed as

$$\overline{M}_g = \frac{1+f}{\sum \left(\frac{\overline{W}}{\overline{M}} \right)_{\text{components}}} \quad (B12a)$$

The resulting equation is

$$\overline{M}_g = \frac{1+f}{0.034522 + 0.035648 f} \quad (B12b)$$

REFERENCES

1. Dugan, J. F., Jr.; Koenig, R. W.; Whitlow, J. B., Jr.; and McAuliffe, T. B.: Power for the Mach 3 SST. *Astron. Aeron.*, vol. 2, no. 9, Sept. 1964, pp. 44-51.
2. McBride, Bonnie J.; Heimel, Sheldon; Ehlers, Janet G.; and Gordon, Sanford: Thermodynamic Properties to 6000⁰ K for 210 Substances Involving the First 18 Elements. NASA SP-3001, 1963.